Proměnné budou **s1**, **s2**, **r** a maximalizovat budeme **r**.

Poloměr nesmí být větší, než vzdálenost středu kružnice (**s1**, **s2**) od jakékoliv úsečky **l**. Z tohoto vztahu získáme podmínky.

Výpočet vzdálenosti:



Trojúhelníky ABC a BST jsou si podobné $\Rightarrow \frac{|AC|}{|AB|} = \frac{|ST|}{|BS|}$

$$\Rightarrow \frac{|AC|}{s_2 - a_i \cdot s_1 - b_i} = \frac{1}{\sqrt{a_i^2 + 1}}$$

$$|AC| = \frac{s_2 - a_i \cdot s_1 - b_i}{\sqrt{a_i^2 + 1}}$$

Hrany mnohoúhelníku rozdělíme na hrany, které mají nejbližší bod ke středu pod středem, a hrany, které mají nejbližší bod nad středem.

Pro hrany pod středem bude podmínka formulovaná takto:

$$\frac{s_2 - a_i \cdot s_1 - b_i}{\sqrt{a_i^2 + 1}} \geq r$$

Pro hrany nad středem, vyjde vzdálenost záporně, podmínka tedy musí být:

$$\frac{s_2 - a_i \cdot s_1 - b_i}{\sqrt{a_i^2 + 1}} \leq -r$$

Formulace lineárního programu:

Pro proměnné: $\boldsymbol{s_1}, \boldsymbol{s_2}, \boldsymbol{r}$

Maximalizujeme: $\boldsymbol{r}$

Za podmínek:

$$\frac{s_2 - a_i \cdot s_1 - b_i}{\sqrt{{a_i}^2 + 1}} \geq r, \text{ pro hrany pod středem}$$

$$\frac{s_2 - a_i \cdot s_1 - b_i}{\sqrt{{a_i}^2 + 1}} \leq -r, \text{ pro hrany nad středem}$$

Kód:

```
param N_above;
param N_below;
param lines_above{1..N_above, 1..2};
param lines_below{1..N_below, 1..2};

var s1;
var s2;
var r;

maximize z: r;

subject to above {i in 1..N_above}:
   (s2 - (lines_above[i, 1] * s1) - lines_above[i, 2]) / sqrt( (lines_above[i, 1]^2) + 1 ) <= -r;

subject to below {j in 1..N_below}:
   (s2 - (lines_below[j, 1] * s1) - lines_below[j, 2]) / sqrt( (lines_below[j, 1]^2) + 1 ) >= r;

data;

param N_above := 5;
param N_below := 3;

param lines_above : 1 2 :=
   1   6 1
   2   1 3.5
   3   -6 35
   4   -1.8 13
   5   -0.25 6;

param lines_below : 1 2 :=
   1   0.05 0
   2   0.8 -3
   3   -1.5 2;

end;
```

Graf příkladu: